

## Controlling Graph Size: Building Thumbnails and GIF Files Using SAS/GRAPH®

Arthur L. Carpenter, Data Explorations, Carlsbad, California  
Richard O. Smith, Data Explorations, Carlsbad, California

### ABSTRACT

GIF is one of the primary file forms used for graphs that are to be displayed on the web. However because of differences in browsers and the need to control the size of the graphic display, the standard GIF device that is defined with SAS/GRAPH is not always appropriate. This paper discusses alternative methods that can be used to create GIF files that will display graphs of various sizes.

Since secondary considerations as to graph contents, options, and control alternatives become necessary when the size of a graphic display is reduced, these issues are also presented.

### KEYWORDS

graphs, GIF, PNG, SAS/GRAPH, resizing, GDEVICE

### INTRODUCTION

One of the primary formats for graphics images that are to be displayed on the web is the GIF (Graphics Interchange Format) file form. Well suited to the web, this file form is used extensively by SAS/GRAPH programmers and there are several DEVICE drivers that will generate GIF files. Schrader (2003) goes into a great deal of detail on how files of images are constructed as well as the different types of files, fonts, and the advantages/disadvantages of each.

The use of these files can become problematic when the size of the image is not the size that is desired. Incorrect size can occur when browser settings change or are not set to the expected resolution. Sometimes web programmers will want to insert the graph into a web page and limited space requires that you provide an image with specific dimensions.


Fortunately SAS/GRAPH provides us with a number of options and tools that allow us to control the size and appearance of the image. These include GOPTIONS that directly relate to size, DEVICE drivers that create images with specific characteristics, and programming statements that control text. Carpenter and Smith (2002) discuss a variety of device drivers and options that can be used to produce images for the Web.

The images shown in this paper are screen captures of displayed GIF images. This is definitely **NOT** the best way to show images in the printed (or PDF) media, however for this paper we wanted to directly show you the issues with GIF files. Also once the images are imported into the word processor sizing information becomes distorted, so you will need to do some experimenting with regards to size.

### CREATING A GIF FILE

One of the first issues that you need to be aware of when creating GIF files is that the resolution is not nearly as good as when you create files that are to be printed (using drivers such as CGM). Usually a higher resolution will not be an issue when the image is displayed on the monitor, but it does play a role as the images are resized.

The graphs in this paper track the values of ozone through time for three stations. A number of options have been employed to make sure that the image is better suited for the GIF format. These are shown in the following example.



The code used to generate this graph is:

```

goptions reset=all;
goptions gsfmode=replace gsfname=gifloc;
goptions device=gif; ❶
filename gifloc "&path\fullsize.gif"; ❶

symbol1 v=none color=blue l=1 w=2 i=join; ❷
symbol2 v=none color=orange l=1 w=2 i=join;
symbol3 v=none color=red l=1 w=2 i=join;

title1 'Ozone Levels Through Time';
title2 a=90 f=swissb 'Parts per Million'; ❸

axis1 label=none
order=(1 to 5)
major=(width=2) ❹
minor=(width=2 n=1)
width=2; ❺
axis2 major=(width=2)
minor=(width=2 n=2)
width=2
order=('01jan88'd to '01jan89'd by quarter);

proc gplot data=sasclass.ca88air;
plot o3*date=station /
vaxis=axis1 ❻
haxis=axis2;
run;
quit;

```


- ❶ Create a GIF file by using the GIF device driver. The file name extension is also GIF.
- ❷ Lines are widened by using the `width=` option.
- ❸ The rotated title uses the SWISSB font to make the letters thicker.
- ❹ Major and minor tick marks are thickened.
- ❺ The axis and lines of the frame are thickened as well.
- ❻ Axis and legend statements are tied to the plot through PLOT statement options.

## RESIZING A GIF IMAGE

There are several GOPTIONS that can be used to resize an image. Two that are commonly used are HSIZE and VSIZE. Lets assume that the user would like to reproduce the above graph with a size of approximately 6X8inches.

The only change to the previous code is:

```
goptions device=gif;  
goptions hsize=8 vsize=6;  
filename gifloc "&path\inches6x8.gif";
```




One of the first things that we notice is that all the fonts that are not explicitly specified are now of poorer quality. This is because unspecified hardware fonts have been substituted. This is easily fixed by using the GOPTION FTEXT=.

```
goptions hsize=8 vsize=6 ftext=swissb;
```

Other than the fact that "inches" have very little meaning in the virtual world, this graph is not too bad. However if you need to make a 'thumbnail' or substantially reduced size image this technique produces graphs with some issues.

```
goptions vsize=1.5 hsize=2 ftext=swissb;
```




The text, even for the larger title, is virtually unreadable. Notice that the width of the lines has not been reduced proportionately and consequently seem to be much thicker. If one takes this image and expands it passed its intended size, it becomes obvious that the resolution of the graph itself has been changed as well. It is not just a matter of size.

## GIF DEVICE DRIVERS

There are several GIF drivers that are specifically designed to allow you to directly control the size of the graph. Each of these controls the size by specifying the number of horizontal and vertical pixels. This device is specified with the GOPTION DEVICE= .

```
goptions device=gif160;
filename gifloc "&path\gif160.gif";
```



Notice that this driver attempts to correct the line width (in the LEGEND) and text size issues noted above. This image still does not contain the detail to allow it to be expanded, but that of course is not the objective. The unreadable text is still distracting, see the section in this paper with the heading "CONTROLLING TEXT ON REDUCED SIZE IMAGES" for more information on how to work with this text.

Other predefined GIF drivers include:

Device Name	Size in Pixels		Approximate size in Inches	
	Vertical	Horizontal	Vertical	Horizontal
GIF	600	800	6.3	8.4
GIF733	550	733	5.8	7.7
GIF570	430	570	4.5	6.0
GIF373	280	373	2.9	3.9
GIF260	195	260	2.1	2.4
GIF160	120	160	1.3	1.7

As a general rule you will probably more satisfied with the results of these drivers over the use of the HSIZE and VSIZE options.

## MODIFYING DEVICE DRIVERS

You can modify and even create your own device drivers with the use of PROC GDEVICE. It is probably easiest to copy and then modify an existing driver, however there are a number of Technical Support documents available that provide step-by-step instructions on how to create new drivers.

You can start by examining current driver definitions by using:

```
proc gdevice;
run;
```

The procedure is interactive and opens with the GDEVICE DIRECTORY window.

Put an X next to the device of interest to show the first of several windows (such as, DETAIL, PARAMETERS, GCOLORS) that specify all aspects of the driver. Most of the attributes are fairly obvious, but it is not always obvious what should be placed in a given field. These windows are fully documented in the SAS OnLine Doc.

## CONTROLLING TEXT ON REDUCED SIZE IMAGES

As you reduce the size of your image the text must be made relatively larger to still be readable. At some point the text can take up all of the image, so this really is not very useful. In fact, as a general rule, when you want a small graph or a thumbnail version of a larger graph the text and lettering becomes less important.

One technique is to make two versions of the graph. The large one with all the appropriate text and the smaller one with all or virtually all text eliminated. After all the small image cannot really be used to convey detail information but rather to indicate a general sense of what is going on in the graph. The following code shows how you could make the small graph shown above, except with the text removed.


```
goptions reset=all;
goptions gsfmode=replace gsfname=gifloc;
goptions device=gif160;
filename gifloc "&path\gif160notext.gif";

symbol1 v=None color=blue l=1 w=2 i=join;
symbol2 v=None color=orange l=1 w=2 i=join;
symbol3 v=None color=red l=1 w=2 i=join;

title1; ❶

axis1 label=None ❷
      order=(1 to 5)
      value=None ❸
      minor=None ❹
      width=2;
axis2 major=(width=2)
      minor=None
      width=2
      value=None
      label=None
      order=('01jan88'd to '01jan89'd by quarter);

proc gplot data=sasclass.ca88air;
plot o3*date=station /
     vaxis=axis1
     haxis=axis2
     nolegend; ❺
run;
quit;
```




- ❶ The titles are turned off. If there had been a footnote it could have been handled in the same way.
- ❷ The label for this axis is turned off.
- ❸ The values for the major tick marks are turned off. The tick marks themselves are not turned off, however we could have done so by using MAJOR=NONE as an option in the AXIS statement.
- ❹ All minor tick marks are turned off.
- ❺ The default LEGEND is turned off by using the NOLEGEND option.

## OTHER DRIVERS FOR THE WEB

The Portable Networks Graphics, PNG, graphic driver has been introduced as a replacement for the GIF file format. This driver has the advantage that it seems to scale well using the HSIZE and VSIZE options and is fairly tolerant of changes in the aspect ratio. The following code was used to make the PNG file shown below.

```
goptions gsfmode=replace gsfname=gifloc;  
goptions device=png;  
goptions vsize=1 hsize=2;  
filename gifloc "&path\png1X2.png";
```



The WEBFRAME device can be used to display a series of graphs with an HTML document. While the HTML document is designed to display the graphs, in fact, the driver creates a series of GIF files. More importantly for this discussion, two GIF files are actually created for each graph; the primary and a thumbnail version. If you need both sizes and the thumbnail is about the right size, you might consider using WEBFRAME to generate the GIF files and then just ignore the HTML file (INDEX.HTML).

## SUMMARY

When displaying images on the web, GIF is one of the most common file forms. There are a number of tools, options, and procedures within SAS/GRAPH that can be used to generate graphs in the size and resolution range that you require. GOPTIONS such as HSIZE and VSIZE are designed to directly affect the size of the image, but may indirectly introduce some complications in the image itself.

When the size of an image is reduced, especially when the resolution is also reduced, it is usually necessary to also adjust the text and symbol sizes used within the image. Often for thumbnail images the text is removed altogether. Text can be removed by redefining text attributes in TITLE, FOOTNOTE, AXIS, and LEGEND statements.

File forms other than GIF, are also now available. Some, such as PNG, offer advantages over the GIF format.

## REFERENCES

Carpenter, Arthur L. and Richard O. Smith, 2002, "ODS and Web Enabled Device Drivers: Displaying and Controlling Large Numbers of Graphs", Proceedings of the 10<sup>th</sup> Annual Western Users of SAS Software, Inc. Users Group Conference, Cary, NC: SAS Institute Inc., pp. 182-189. Also in the Proceedings of the Pharmaceutical SAS<sup>®</sup> Users Group Conference (2003), Cary, NC: SAS Institute Inc., paper TT026.

Schrader, Dawn, "Exporting SAS/GRAPH<sup>®</sup> Output: Concepts and Ideas", Proceedings of the Twenty-Eighth Annual SAS<sup>®</sup> Users Group International Conference, Cary, NC: SAS Institute Inc., paper 139. Also in the Proceedings of the Pharmaceutical SAS<sup>®</sup> Users Group Conference, Cary, NC: SAS Institute Inc. (2003), paper SAS130.

## ABOUT THE AUTHORS

Richard Smith and Art Carpenter are senior partners at Data Explorations. Data Explorations, a SAS Alliance Affiliate Member<sup>™</sup>, provides data management, analyses, and SAS programming services nationwide.

### Richard O. Smith

Richard Smith has a Masters in Biology/Ecology and has provided complete data management and analysis services for numerous environmental research projects as a senior biologist, SAS programmer, and project manager. He has also provided programming and management services for the health related industries. He has been using SAS extensively since 1981 and is a SAS Certified Professional<sup>™</sup>.

### Arthur L. Carpenter

Art Carpenter is a SAS Certified Professional<sup>™</sup>. His publications list includes three books on SAS topics (*Annotate: Simply the Basics*, *Quick Results with SAS/GRAPH<sup>®</sup> Software*, and *Carpenter's Complete Guide to the SAS<sup>®</sup> Macro Language*), two chapters in *Reporting from the Field*, and numerous papers and posters presented at various user group conferences. Art has been using SAS since 1976 and has served in a variety of positions in user groups at the local, regional, and national level.

## AUTHOR CONTACTS

Data Explorations  
2270 Camino Vida Roble, Suite L  
Carlsbad, CA 92009

Richard O. Smith   Arthur L. Carpenter  
(760) 438-1336                      (907) 865-9163  
[RSmith@SciX.com](mailto:RSmith@SciX.com)                      [Art@Caloxy.com](mailto:Art@Caloxy.com)



## TRADEMARK INFORMATION

SAS, SAS Certified Professional, SAS Alliance Affiliate, and SAS products are registered trademarks of SAS Institute, Inc. in the USA and other countries. ® indicates USA registration.